

2022

Monte Carlo and Bootstrapping Carry Trade Simulations in Excel

Tom Arnold

University of Richmond, tarnold@richmond.edu

C. Mitchell Conover

University of Richmond, mconover@richmonde.edu

Joseph Farizo

jfarizo@richmond.edu

Follow this and additional works at: <https://scholarship.richmond.edu/finance-faculty-publications>



Part of the [Finance and Financial Management Commons](#)

Recommended Citation

Arnold, Tom, C. Mitchell Conover, and Joseph Farizo. "Monte Carlo and Bootstrapping Carry Trade Simulations in Excel," University of Richmond Robins School of Business, (2022): 1-18.

This Article is brought to you for free and open access by the Finance at UR Scholarship Repository. It has been accepted for inclusion in Finance Faculty Publications by an authorized administrator of UR Scholarship Repository. For more information, please contact scholarshiprepository@richmond.edu.

Monte Carlo and Bootstrapping Carry Trade Simulations in Excel

Tom Arnold, CFA, CIPM
The Robins School of Business
University of Richmond
Department of Finance
102 UR Drive
Richmond, VA 23173
tarnold@richmond.edu
O: 804-287-6399
F: 804-289-8878

C. Mitchell Conover, CAIA, CFA, CIPM
The Robins School of Business
University of Richmond
Department of Finance
102 UR Drive
Richmond, VA 23173
mconover@richmond.edu
O: 804-287-1921
F: 804-289-8878

Joseph Farizo
The Robins School of Business
Department of Finance
102 UR Drive
University of Richmond, VA 23173
O: 804-289-8565
F: 804-289-8878
jfarizo@richmond.edu

Please do not cite without permission

August 11, 2022

Monte Carlo and Bootstrapping Carry Trade Simulations in Excel

In a currency carry trade, an investor borrows money in a low interest rate currency and invests in a high interest rate currency. The trade is profitable if the future exchange rate does not adjust to the interest rate differential. After downloading exchange rate data, a Monte Carlo simulation of a carry trade is performed in Excel based on a normal distribution and the data's mean and standard deviation. A bootstrapping carry trade simulation exercise is also generated by randomly selecting observations from the historical data. In contrast to the Monte Carlo simulation, the bootstrapping exercise preserves the skewness within the historical data. A carry trade simulation can generate student interest and strengthen student understanding of Excel modeling, Investments, International Finance, and Statistics. The assignment can also be used to illustrate uncovered interest rate parity and forward rate bias.

INTRODUCTION

In currency markets, the carry trade consists of borrowing a low interest rate currency, converting the currency to another at the spot rate, and investing in the second, higher interest rate currency for typically a year or less. When the trade is closed, the investment is liquidated, the second currency is converted at the then spot rate to the first currency, and the loan is paid off. For example, if a trader can borrow at 3% for one year in the US and invest at 8% in Brazil, the trader will borrow dollars and convert to the Brazilian currency (the real) at trade initiation. After investing in the Brazilian money market at 8% for one year, the real is converted to dollars at the then spot rate and the loan is repaid. If the real has not changed in value, the trader's profit is approximately the difference in interest rates of 5%. If the real appreciates, the profit is greater than 5%. If the real depreciates but not more than 5%, the profit is 5% minus the depreciation. If the real depreciates by more than 5%, the trader's losses are equal to the depreciation net of 5%. Consequently, a carry trade is risky and not an arbitrage opportunity.

In the first section, the carry trade and uncovered interest rate parity are discussed with the following section providing information on empirical investigations. Next, a numerical example is provided followed by a Monte Carlo simulation and a Bootstrapping simulation in Excel.

THE CARRY TRADE AND UNCOVERED INTEREST RATE PARITY

The carry trade's risk results from the potential change in the exchange rate, which is unknown at the initiation of the trade. The breakeven future exchange rate where carry

trade profits are equal to zero is given by the Uncovered Interest Rate Parity (UIRP) relationship from International Finance, stated as:

$$E(S_1) / S_0 = (1 + R_A) \div (1 + R_B) \quad (1)$$

where S_0 is the spot rate at time zero, $E(S_1)$ is the expected spot rate in one period, and R is the nominal interest rate in country A and country B. UIRP relates the spot rate in one period to interest rates in each country and the current spot rate. In equilibrium, UIRP implies that carry trade profits will equal zero because the future spot rate would fully reflect the nominal interest rate differential between countries.

To determine the zero-profit future spot rate in the carry trade, the UIRP relationship is rearranged as:

$$E(S_1) = S_0 \times [(1 + R_A) \div (1 + R_B)] \quad (2)$$

In our example, Brazilian real strength being greater than the breakeven spot rate indicates carry trade profits.

UIRP is one of the most closely studied parity conditions in international finance. Under UIRP, real interest rates are assumed to be equal throughout the world and the only difference in nominal interest rates will be from inflation differentials. As a result, currencies with higher inflation, e.g., emerging markets, will depreciate by approximately the interest rate differential such that carry trades will not earn profits.

In contrast, if UIRP does not hold, then average returns will be non-zero. The strongest evidence against UIRP is provided if the high interest rate currency appreciates. Contrary to the UIRP theory, it has been found that in the short-run, high interest rate currencies actually do appreciate. This empirical observation is referred to as forward rate bias, and is discussed in class as a counterargument to international parity conditions.

Given the short-term nature of the project's trade (one quarter), students often find support for forward rate bias.

Thus, while UIRP is typically used in International Finance classes to examine a theoretical basis for exchange rate determination, it also has practical use in the carry trade, which generates student interest and reinforces fundamental concepts in international finance.

CARRY TRADE PROFITS AS A RISK PREMIUM

The support for UIRP is based on long term data. However, previous research has found that high interest rate currencies actually appreciate and carry trades can generate profits over short time horizons. For example, Lothian and Wu (2011) find that UIRP holds over the long term and for large interest rate differentials but that the relationship is weaker in the short run. Using recent data from 1976 to 2009, Burnside et. al. (2011) find average returns of 4.3% annually, after transactions costs.¹ Using a longer span of data from 1900 to 2011, Dimson, Marsh, and Staunton (2012) report more modest average carry trade returns of 1.1% annually. Carry trade returns appear to reflect a risk premium as Czaronis, Pamir and Turkington (2019) find that riskier pairs generate higher returns.

As implemented, the carry trade is often long an emerging market currency, where interest rates are typically higher than in developed markets. As such, carry trade profits are often attributed to the "peso problem" where the potential depreciation of risky emerging market currencies can result in carry trade losses.² Losses can be especially acute

¹ They attribute profits to the peso problem.

² The terminology "peso problem" is attributed to Milton Friedman's observation that although the Mexican peso was fixed against the dollar in the 1970s, the much higher peso interest rate implied that the peso would eventually devalue, which it subsequently did by 46% in 1976. See Sill (2000) for additional discussion.

when global instability arises, since developed market currencies often surge in value against emerging market currencies. Thus, it is often the case that the distribution of carry trade profits is negatively skewed, reflecting occasional large losses.

Although these issues are discussed in class and tested, student understanding and long-term recall can be enhanced through the use of this modeling project. Students are instructed that if they want to work in the currency markets, it is imperative that they understand the carry trade, given its importance.

CARRY TRADE DATA AND CALCULATIONS

In this section, the carry trade is demonstrated in a step-by-step manner with the following two sections introducing two different simulation techniques for analyzing the risk associated with the carry trade. The first is a Monte Carlo simulation in which the future 3-month exchange rate is generated based on a normal distribution and the mean and standard deviation from the historical data. The second simulation exercise uses a bootstrapping technique in which the simulation randomly selects 3-month changes in the exchange rate from the historical data.

To simulate the carry trade, ten years of exchange rate data are downloaded from Yahoo Finance (www.finance.yahoo.com) along with government treasury rates from Moody's Analytics (www.economy.com/indicators/treasury-bills-over-31-days). These sources are used because they are readily available, however, other internet sources exist and proprietary sources, such as, Bloomberg, are also potentially available.

Viewing an exchange rate as a “direct quote” means a single unit of a foreign currency “f” is valued using a domestic currency “d.” For example, one unit of “f” may currently cost 1.20 units of “d”.

$$1.20 \text{ d} : 1.00 \text{ f} \quad (3)$$

Let the domestic risk-free rate be 3% APR (R_{F-DOM}) and let the foreign risk-free rate be 8% APR (R_{F-FOR}). Three months from today, if UIRP holds both currencies will adjust by each earning its respective risk-free rate:

$$1.20 \text{ d} \times [1 + (R_{F-DOM} \div 4)] : 1.00 \text{ f} \times [1 + (R_{F-FOR} \div 4)] \quad (4)$$

Calculate the 3-month expected quote as a direct quote:

$$1.20 \text{ d} \times \{[1 + (R_{F-DOM} \div 4)] \div [1 + (R_{F-FOR} \div 4)]\} : 1.00 \text{ f} \quad (5)$$

Introduce the risk-free rates of 3% and 8%:

$$1.20 \text{ d} \times \{[1 + (3\% \div 4)] \div [1 + (8\% \div 4)]\} : 1.00 \text{ f} \quad (6)$$

$$1.18529 \text{ d} : 1.00 \text{ f} \quad (7)$$

In the theory of UIRP, the future 3-month exchange rate should be 1.18429, however, what happens over the next three months will change depending on market conditions. The carry trade will be lucrative if the future exchange rate is above 1.18529 d : 1.00 f.

The carry trade works in this manner:

1. Borrow money in the currency that has the lower interest rate (i.e. the domestic currency in this example) for three months. Assume borrowing 10,000 d, which means in three months, you will owe 10,075 d (= 10,000 × {[1 + (3% ÷ 4)]}).
2. Convert the borrowed money into the currency that has the higher interest rate (i.e. the foreign currency in this example). 10,000 d becomes 8,333.3333 f (= 10,000 d ÷ 1.20)

3. Invest the converted borrowed money at the higher interest rate for three months. After three months it will be 8,500 f ($= 8,333.3333 \times [1 + (8\% \div 4)]$)
4. Convert the money invested in the higher interest rate back to the currency with the lower interest rate and repay the loan.

If the exchange rate in three months is 1.18529 d : 1.00 f as demonstrated in equations (4-7), the 8,500 f converts back to 10,075 d ($= 8,500 \times 1.18529$, note, there is some rounding error) and the carry trade does not have a profit nor a loss. However, if the foreign currency appreciates above 1.18529 d : 1.00 f, say 1.19 d : 1:00 f, the carry trade will generate a profit. The 8,500 f converts to 10,115 d ($= 8,500 \times 1.19$) and generates a profit of 40 d after the loan is repaid ($= 10,115 \text{ d} - 10,075 \text{ d}$). Conversely, an exchange rate below 1.18529 d : 1.00 f will result in a loss. Consequently, the investment strategy behind a carry trade is that the future exchange rate does not eliminate the interest rate differential (or spread).

To provide students an example of how an exchange rate may change and create a profit or loss through a carry trade, a Monte Carlo simulation is demonstrated in the next section using the U.S. dollar (low interest rate environment) and the Brazilian real (high interest rate environment). It should be noted that the exchange rate data in the next section are quoted as indirect exchange rates. An indirect exchange rate is expressed as the amount of foreign currency (i.e. Brazilian real) needed to be exchanged for one unit of domestic currency (i.e. U.S. dollar). If we divide both sides of equation (3) by 1.20, the indirect exchange rate is produced.

$$1.00 \text{ d} : 0.8333 \text{ f} \tag{8}$$

MONTE CARLO SIMULATION

To generate the parameters for the simulation, 10 years of monthly exchange rate data is downloaded from Yahoo Finance (12/01/2011 to 12/01/2021, using the symbol: BRL=X, 121 observations). The data is imported into Excel and is sorted so that the most recent data is displayed at the top of the column in the spreadsheet. The monthly data is converted to provide the quarterly percentage change in the exchange rate (similar to calculating a quarterly return with monthly security prices). Each quarterly percentage change is given an index number between 1 and 118 with 1 being the most recent quarterly percentage change (see Table 1). Note: the exchange rate data are expressed as indirect exchange rate quotes for U.S. dollars (i.e. units of Real to one U.S. dollar).

Table 1: Excel Spreadsheet of 10 Years of Quarterly Changes in the BRL-to-USD Exchange Rate (DATA SORTED TAB)

	A	B	C	D	E	F	G	H	I	J
1	Date	Open	High	Low	Close	Adj Close	Volume		INDEX	QTR CHG
2	12/1/2021	5.62	5.69	5.58	5.64	5.64	0		1	4.23%
3	11/1/2021	5.64	5.70	5.39	5.60	5.60	0		2	8.06%
4	10/1/2021	5.44	5.75	5.36	5.63	5.63	0		3	8.29%
5	9/1/22021	5.15	5.46	5.13	5.41	5.41	0		4	9.27%
6	8/1/2021	5.20	5.47	5.11	5.18	5.18	0		5	-1.05%
7	7/1/2021	4.97	5.31	4.95	5.20	5.20	0		6	-2.51%
8	6/1/2021	5.22	5.22	4.89	4.96	4.96	0		7	-14.15%
9	5/1/2021	5.43	5.48	5.20	5.24	5.24	0		8	-6.42%

	K	L	M	N	O	P	Q	R	S	T
1						TRIAL:	BASE-SIM1	BASE-SIM2		
2		BRL-XRATE:	5.643322			1	10.4470%	-3.2122%		
3		BRL-RATE:	11.36%	APR		2	1.0856%	9.9995%		
4		USD-RATE:	0.05%	APR		3	10.4867%	-10.0209%		
5						4	5.4603%	-0.5538%		
6		Mean:	3.3089%	QTR		5	-5.6487%	-9.1137%		
7		Volatility:	8.3043%	QTR		6	4.9385%	16.8271%		
8						7	-7.4712%	8.7229%		
9						8	10.7997%	5.7820%		

Columns A through G are sorted data from Yahoo Finance (www.finance.yahoo.com)
 Cell I2 is set to 1, Cell I3: = 1 + I2, and copy downward until the index reaches a value of 118
 Cell J2: =(E2 - E5)/E5, this calculates the quarterly percentage change in the exchange rate using the "Close" value, copy the cell downward to fill out the 118 indexed values pertaining to the 121 historical observations.

Cell M1: =E2, current USD-BRL exchange rate.

Cell M2: current BRL risk-free rate from Moody's Analytics (www.economy.com/indicators/treasury-bills-over-31-days)

Cell M3: current USD risk-free rate from Moody's Analytics (www.economy.com/indicators/treasury-bills-over-31-days)

NOTE: The information below will be applied later in the paper as the two different simulation techniques are introduced.

Cell M6: =AVERAGE(J2:J119), calculates the historic mean of the percentage quarterly changes in the exchange rate.

Cell M7: =STDEV(J2:J119), calculates the historic standard deviation of the percentage quarterly changes in the exchange rate.

Cell P2 is set to 1, Cell P3: = 1 + P2, and copy downward until it reaches a value of 500

Cell Q2: =NORMINV(RAND(),\$M\$6,\$M\$7), this cell is then copied to generate 500 trials. Next, copy the results and paste the results over the column of trials as “values.” The purpose of this column is to set up a “base case” for the Monte Carlo simulation.

Cell R2: =VLOOKUP(RANDBETWEEN(1,118),\$I\$2:\$J\$119),2), this cell is then copied to generate 500 trials. Next, copy the results and paste the results over the column of trials as “values.” The purpose of this column is to set up a “base case” for the Bootstrapping simulation.

A copy of this file is available at: <https://scholarship.richmond.edu/finance-faculty-publications/50/>

To generate a Monte Carlo simulation, it is assumed that the quarterly percentage change in the current exchange rate is selected from a normal distribution with a mean of 3.3089% (cell M6 in Table 1) and a standard deviation (or volatility) of 8.3043% (cell M7 from Table 1) based on the mean and standard deviation computed from the historical data. In a given “trial” within the simulation (500 trials will be performed), a value is randomly taken from the normal distribution and then applied to the current exchange rate of 5.5433 USD-BRL. The Excel function to select the random variable is: =NORMINV(RAND(), 3.3089%, 8.3043%), where the =NORMINV(X, mean, standard deviation) function selects a value associated with its probability within the cumulative normal distribution with a mean of 3.3089% and a standard deviation of 8.3043% based on a number X which is between 0 and 1. The RAND() function, selects a random number between 0 and 1 and will keep generating new random values anytime the spreadsheet re-computes (this feature can be turned off if desired: File/Options/Formulas/switch to “Manual” under “Calculation options”). Once the simulation is completed, pressing the “F9” key will generate a new set of random variables.

In Table 2, the Monte Carlo simulation is generated in Excel for a 3-month carry trade:

1. \$1,000.00 is borrowed at the current U.S. risk-free rate of 0.05% APR for three months (cell B1 in Table 2). In three months, \$1,000.1250 must be repaid (cell B2 in Table 2).

2. The borrowed money is converted to Brazilian real at the current exchange rate of 5.6433 BRL-USD. This equates to having 5,643.3220 Brazilian real (cell B3 in Table 2).
3. The converted borrowed money is invested at the current Brazilian risk-free rate of 11.36% APR. In three months, it will be worth 5,803.5923 Brazilian real (column D in Table 2).
4. In a given simulation trial, an exchange rate is produced that is the simulated spot exchange rate three months in the future (column B in Table 2 is the simulated change in the exchange rate and column C in Table 2 is the simulated 3-month exchange rate). When generated, the 15,803.5923 Brazilian real (column D in Table 2) are converted back to U.S. dollars (column E in Table 2), the loan is repaid, and it is determined if the carry trade produced a profit or loss (column F in Table 2 is the profit/loss and column G in Table 2 is the 3-month return).

Table 2: Excel Spreadsheet for Monte Carlo Carry Trade Simulation (SIM-1 TAB)

	A	B	C	D	E	F	G
1	BORROW-US:	\$ 1,000.0000			BRL-XRATE:	5.6433	BRL-USD
2	OWE (90 days):	\$ 1,000.1250			BRL-RATE:	11.36%	APR
3	CONVERT TO BRL:	5643.3220			USD-RATE:	0.05%	APR
4	QUOTE-SWITCH:	1	"0" for direct quote with lower interest currency, "1" for indirect quote with lower interest currency				
5	SIM-SWITCH:	0	"0" resets to base values, "1" allows for new simulation values				
6							
7	TRIAL:	XRATE-CHG:	XRATE-FUT:	INVEST-BRL:	CONVERT-USD:	PROFIT:	RETURN:
8	1	10.4470%	6.2329	5803.5923	\$931.12	-\$69.00	-6.90%
9	2	1.0856%	5.7046	5803.5923	\$1,017.36	\$17.23	1.72%
10	3	10.4867%	6.2351	5803.5923	\$930.79	-\$69.33	-6.93%
11	4	5.4603%	5.9515	5803.5923	\$975.15	-\$24.97	-2.50%
12	5	-5.6487%	5.3245	5803.5923	\$1,089.97	\$89.84	8.98%
13	6	4.9385%	5.9220	5803.5923	\$980.00	-\$20.12	-2.01%

	H	I	J	K	L	M	N
1		Mean:	3.3089%	QTR			
2		Volatility:	8.3043%	QTR			
3							
4							
5							
6							
7			PROFIT:	RETURN:			

8		AVG:	\$8.42	0.842%		
9		MAX:	\$307.44	30.744%		
10		MIN:	-\$189.32	-18.932%		
11		MEDIAN:	\$0.92	0.092%		
12		GAIN:	253			
13		LOSS:	247			

Cell B2: =B1*(1 + F3/4), determines how much USD will be owed in three months
Cell B3: =IF(B4=0, B1/F1,B1*F1), converts borrowed \$1,000.00 to BRL, depending on direct or indirect quote input in cell B4

Cell F1: ='DATA-SORTED'!M2, refers back to spreadsheet tab in Table 1, alternatively can type in 5.4633
Cell F2: ='DATA-SORTED'!M3, refers back to spreadsheet tab in Table 1, alternatively can type in 11.36%
Cell F3: ='DATA-SORTED'!M4, refers back to spreadsheet tab in Table 1, alternatively can type in 0.05%

Cell J1: ='DATA-SORTED'!M6, refers back to spreadsheet tab in Table 1, mean based on historic data
Cell J2: ='DATA-SORTED'!M7, refers back to spreadsheet tab in Table 1, standard deviation based on historic data

QUOTE-SWITCH: Cell B4 is zero or a non-zero value, zero will provide a direct quote while a non-zero value will provide an indirect quote.

SIM-SWITCH: Cell B5 is zero or a non-zero value, zero will provide a “base case” Monte Carlo simulation based on the “set” random values from Column Q in Table 1 (i.e. spreadsheet tab ‘DATA-SORTED’). This is useful for introducing the simulation technique because the values do not re-compute continuously with the spreadsheet.

SIMULATION PORTION:

Cell A8 is set to 1, Cell A9: = 1 + A8, and copy downward until it reaches a value of 500 (i.e. 500 simulation trials)

Cell B8: =IF(\$B\$5=0,'DATA-SORTED'!Q2,NORMINV(RAND(),\$J\$1,\$J\$2)), the if-statement allows for the “base case” simulation if cell B5 is zero or if B5 is not zero, a value is randomly selected from a normal distribution with mean of 3.3089% (cell J1) and a standard deviation or volatility of 8.3043% (cell J2). Copy this cell down the column for 500 trials.

Cell C8: =\$F\$1*(1+B8), calculates the simulated 3-month exchange rate
Cell D8: =\$B\$3*(1 + \$F\$2/4), calculates the value of the 3-month investment into Brazilian real
Cell E8: =IF(\$B\$4=0,D8*C8,D8/C8), the if-statement allows for a direct quote (if zero) and an indirect quote (if non-zero) based on the simulated exchange rate
Cell F8: =E8 – \$B\$2, determines the profit or loss with the 3-month investment in terms of U.S. dollars
Cell G8: =F8/\$B\$1, coverts the profit or loss into a 3-month return


Cell J8: =AVERAGE(F8:F507), computes the average profit/loss over the 500 trials
Cell K8: =J8/B1, converts the average 3-month profit or loss to a return
Cell J9: =MAX(F8:F507), the max profit in dollars of the simulation
Cell K9: =MAX(G8:G507), the max profit in return of the simulation
Cell J10: =MIN(F8:F507), the min profit in dollars of the simulation
Cell K10: =MIN(G8:G507), the min profit in return of the simulation
Cell J11: =MEDIAN(F8:F507), the median profit in dollars of the simulation
Cell K11: =MEDIAN(G8:G507), the median profit in return of the simulation

Cell J12: =COUNTIF(F8:F507,“>0”), counts the number of trials in which there is a profit
Cell J13: = 500 – J12, counts the number of trials in which there is zero profit or a loss

A copy of this file is available at: <https://scholarship.richmond.edu/finance-faculty-publications/50/>

The student can be shown that the (base case) simulation demonstrates that the carry trade will most likely be profitable based on past changes in the exchange rate. However, there is no guarantee that a profitable trade will occur in the future. One can also do further analysis on the simulated data. Figure 1 displays a histogram of the return from the “base case” Monte Carlo simulation.

Figure 1: Base Case Monte Carlo Simulation Return Histogram (SIM 1 TAB)



The histogram provides an excellent basis for the discussion of the carry trade’s risk and UIRP. Although the mean return in this three-month carry trade is positive, one can see that there are many occurrences of losses in the histogram, with some of the losses being quite large, as with the “peso problem.” The distribution provides a visual demonstration of carry trade risk. Furthermore, if UIRP were to always hold, then instead of predominantly positive returns, returns will always be zero. Recall that UIRP states that real rates are the same throughout the world and that high nominal interest rate currencies have higher inflation, which indicates they will depreciate by approximately the size of the interest rate differential. The histogram can generate a discussion of the “peso problem” and the real-world relevance of UIRP and forward rate bias.

By changing cell B5 to “1”, new simulations are produced by pressing the “F9” key. When presenting to students, it should be noted that each new run of 500 simulations will produce similar results, but never exactly the same results. In fact, it can be discussed how many trials within a simulation are needed for simulation results to become more consistent with each other. Further, one can discuss how an analysis can be biased by using a particular set of trials rather than trying to be objective.

BOOTSTRAPPING SIMULATION

A bootstrapping simulation differs from a Monte Carlo simulation in that a random change in the exchange rate is not from a probability distribution, but rather from a random selection of historical changes in the exchange rate. Bootstrapping preserves more of what actually occurred historically with the exchange rate and assumes that each historical event has the same probability of occurring in the future. This does not necessarily make bootstrapping a better method than a Monte Carlo simulation, but it is an alternative method that is used. In many cases, especially in emerging markets, exchange rates are not normally distributed in which case bootstrapping would be preferred. In most simulations, bootstrapping will better preserve and demonstrate the fat tails present in many emerging market distributions.

Table 3: Excel Spreadsheet for Bootstrapping Carry Trade Simulation (SIM-2 TAB)

	A	B	C	D	E	F	G
1	BORROW-US:	\$ 1,000.0000			BRL-XRATE:	5.6433	BRL-USD
2	OWE (90 days):	\$ 1,000.1250			BRL-RATE:	11.36%	APR
3	CONVERT TO BRL:	5643.3220			USD-RATE:	0.05%	APR
4	QUOTE-SWITCH:	1	"0" for direct quote with lower interest currency, "1" for indirect quote with lower interest currency				
5	SIM SWITCH:	0	"0" resets to base values, "1" allows for new simulation values				
6							
7	TRIAL:	XRATE-CHG:	XRATE-FUT:	INVEST-BRL:	CONVERT-USD:	PROFIT:	RETURN:
8	1	-3.2122%	5.4620	5803.5923	\$1,062.53	\$62.41	6.24%
9	2	9.9995%	6.2076	5803.5923	\$934.91	-\$65.21	-6.52%
10	3	-10.0209%	5.0778	5803.5923	\$1,142.93	\$142.81	14.28%
11	4	-0.5538%	5.6121	5803.5923	\$1,034.13	\$34.00	3.40%
12	5	-9.1137%	5.1290	5803.5923	\$1,131.52	\$131.40	13.14%

13	6	16.8271%	6.5929	5803.5923	\$880.28	-\$119.85	-11.98%
----	---	----------	--------	-----------	----------	-----------	---------

	H	I	J	K	L	M	N
1							
2							
3							
4							
5							
6							
7			PROFIT:	RETURN:			
8		AVG:	\$12.12	1.212%			
9		MAX:	\$218.27	21.827%			
10		MIN:	-\$210.00	-21.000%			
11		MEDIAN:	\$16.40	1.640%			
12		GAIN:	277				
13		LOSS:	223				

Cell B2: =B1*(1 + F3/4), determines how much USD will be owed in three months
Cell B3: =IF(B4=0, B1/F1,B1*F1), converts borrowed \$1,000.00 to BRL, depending on direct or indirect quote input in cell B4

Cell F1: ='DATA-SORTED'!M2, refers back to spreadsheet tab in Table 1, alternatively can type in 5.4633
Cell F2: ='DATA-SORTED'!M3, refers back to spreadsheet tab in Table 1, alternatively can type in 11.36%
Cell F3: ='DATA-SORTED'!M4, refers back to spreadsheet tab in Table 1, alternatively can type in 0.05%

Cell J1: ='DATA-SORTED'!M6, refers back to spreadsheet tab in Table 1, mean based on historic data
Cell J2: ='DATA-SORTED'!M7, refers back to spreadsheet tab in Table 1, standard deviation based on historic data

QUOTE-SWITCH: Cell B4 is zero or a non-zero value, zero will provide a direct quote while a non-zero value will provide an indirect quote.

SIM-SWITCH: Cell B5 is zero or a non-zero value, zero will provide a "base case" Monte Carlo simulation based on the "set" random values from Column Q in Table 1 (i.e. spreadsheet tab 'DATA-SORTED'). This is useful for introducing the simulation technique because the values do not re-compute continuously with the spreadsheet.

SIMULATION PORTION:

Cell A8 is set to 1, Cell A9: = 1 + A8, and copy downward until it reaches a value of 500 (i.e. 500 simulation trials)

Cell B8: =IF(\$B\$5=0,'DATA-SORTED'!R2,VLOOKUP(RANDBETWEEN(1,118),'DATA-SORTED'!\$I\$2:\$J\$119,2)), the if-statement allows for the "base case" simulation if cell B5 is zero or if B5 is not zero, a value is randomly selected from the 118 historical 3-month changes in the exchange rates from the past ten year. Copy this cell down the column for 500 trials.

Cell C8: =\$F\$1*(1+B8), calculates the simulated 3-month exchange rate
Cell D8: =\$B\$3*(1 + \$F\$2/4), calculates the value of the 3-month investment into Brazilian real
Cell E8: =IF(\$B\$4=0,D8*C8,D8/C8), the if-statement allows for a direct quote (if zero) and an indirect quote (if non-zero) based on the simulated exchange rate
Cell F8: =E8 - \$B\$2, determines the profit or loss with the 3-month investment in terms of U.S. dollars
Cell G8: =F8/\$B\$1, converts the profit or loss into a 3-month return


Cell J8: =AVERAGE(F8:F507), computes the average profit/loss over the 500 trials
Cell K8: =J8/B1, converts the average 3-month profit or loss to a return
Cell J9: =MAX(F8:F507), the max profit in dollars of the simulation
Cell K9: =MAX(G8:G507), the max profit in return of the simulation
Cell J10: =MIN(F8:F507), the min profit in dollars of the simulation
Cell K10: =MIN(G8:G507), the min profit in return of the simulation
Cell J11: =MEDIAN(F8:F507), the median profit in dollars of the simulation
Cell K11: =MEDIAN(G8:G507), the median profit in return of the simulation

Cell J12: =COUNTIF(F8:F507,">0"), counts the number of trials in which there is a profit
Cell J13: = 500 - J12, counts the number of trials in which there is zero profit or a loss

A copy of this file is available at: <https://scholarship.richmond.edu/finance-faculty-publications/50/>

The (base case) bootstrapping simulation demonstrates the carry trade to have even more instances of profitability when compared to the base case Monte Carlo simulation. Further, it is possible to have multiple trials with the same value because there are only 118 possible choices for 500 trials. Students should be made aware of the differences between the two techniques. A histogram of the return is presented in Figure 2, displaying how the historical data used within the bootstrapping method is more skewed than the normally distributed Monte Carlo simulation. As with the Monte Carlo simulation, the histogram in Figure 2 demonstrates the risk of the carry trade and can be used as a basis for the discussion of UIRP and forward rate bias. Additionally, students can visualize that carry trade profits may not conform to a normal distribution, implying that the standard deviation is not always the best or a complete measure of carry trade risk.

Figure 2: Base Case Monte Carlo Simulation Return Histogram (SIM 2 TAB)



Similar to the Monte Carlo simulation, changing cell B5 to “1”, will allow new simulations to be generated when pressing the “F9” key. Further, similar to the Monte Carlo simulation, when presenting to students, it should be noted that each new run of 500 simulations will produce similar results, but never exactly the same results.

Having more trials becomes an interesting question because of the limitations of the sample from which trials are randomly sampled (i.e. there are only 118 choices). Further, one can consider only the most recent five years of data to generate a smaller, but more recent sample to select from randomly. And, as with the Monte Carlo simulation, one can discuss, biasing an analysis toward a particular set of trials rather than trying to be objective.

CONCLUSION

Undergraduate students typically prefer real world applications, rather than theory such as international parity conditions. Furthermore, simulation tends to be intimidating and is often considered only when using specialized software. The carry trade is a widely practiced short-term trade in Investments and International Finance. In this paper, the principles of Monte Carlo and bootstrapping simulations are presented within the context of a carry trade. Excel provides a “visual” platform to display how a simulation is performed and the information generated which informs a trading decision. In our example, one determines the risk of a carry trade based on ten years of historical data.

Further, the instructor can discuss different issues in regard to simulation, e.g., the appropriate number of trials, objectivity, etc. Such a discussion would be appropriate in a fintech setting or even an advanced finance class that is computationally oriented.

However, the simulation models are also useful to illustrate the riskiness of a carry trade whether students are required to perform a simulation or not. Just by itself, the presentation of the base case simulations demonstrates how the carry trade can be profitable, which illustrates fundamental concepts in International Finance. From this assignment, students gain greater understanding of Excel modeling, the carry trade, uncovered interest rate parity, and forward rate bias.

REFERENCES

Burnside, Craig, Martin Eichenbaum, Isaac Kleshchelski, and Sergio Rebelo. 2011. "Do peso problems explain the returns to the carry trade?" *The Review of Financial Studies*, 24:3, 853-891.

Czasonis, Megan, Baykan Pamir, and David Turkington. 2019. "Carry On." *The Journal of Alternative Investments*, 22:2, 100-111.

Dimson, Elroy, Paul Marsh, and Mike Staunton. 2012. "Currency matters." in *Credit Suisse Global Investment Returns Yearbook*, Zurich: Credit Suisse Research Institute.

Lothian, James R. and Liuren Wu. 2011. "Uncovered interest-rate parity over the past two centuries." *Journal of International Money and Finance*, 30:3, 448-473.

Sill, Keith. 2000. "Understanding asset values: stock prices, exchange rates, and the 'Peso Problem'" *Business Review*, Federal Reserve Bank of Philadelphia.